

Global Macro for Master Tracker

Zhuo Chen, Ted Lystig, Martha Cao and Tuan Nguyen
BridgeBio Pharma, LLC

ABSTRACT

In a clinical study, it is important to create and maintain the master tracker for SDTM, ADaM, Tables Listings, Figures (TLF) development and validation for each delivery. Normally a clinical study has hundreds or even thousands of TLF outputs, it is very time-consuming if manually creating it for each delivery.


This paper describes a global Macro which can automate the master tracker efforts. It also check different issues and flag them in the master tracker, therefore help team to guarantee accuracy, gain efficiency and avoid human errors.

INTRODUCTION

This global macro only needs one macro variable, the location of the deliverables, then it will automatically search each SDTM, ADaM, Tables, Figures and Listings subfolder for all SAS programs, validation programs and outputs, then generate a master tracker in EXCEL format. This global macro can also find potential issues during SDTM, ADaM and TLF development and validation process.

STUDY FOLDER STRUCTURE

Companies normally have standard folder structures for clinical studies. Display 1 is one of the examples. “develop” folder contains developers SAS programs. “validation” folder contains validators SAS programs and LST file. “data” folder contains SDTM or ADaM data set. “output” folder contains TLF outputs.



Display 1. Example of Study Folder Structure

If your company's standard folder structure is different, you can slightly update key words in Global Macro. Later, we will use another example to describe how to implement it based on your company's standard folder structure.

AN EXAMPLE OF MASTER TRACKER GENERATED BY THE GLOBAL MACRO

Output 1 shows an example of MasterTracker.xlsx automatically generated by the Global Macro %Tracker. It contains sdtm, adam, tables, figures and listings tab. See details in table tab below. Other tabs contain the same columns but unique informations for sdtm, adam, figures and listings.

	A	B	C	D	E	F	G	H	I	J	K
	Develop Program Name	Develop Program Date-time	Developer	Validation Program Name	Validation Program Date-time	Validator	Validation LST File Name	Validation LST Date-time	Output Name	Output Date-time	Issue(s)
1	t-14-01-01.sas	12/13/2023 11:13AM	\JChen	v-t-14-01-01.sas	03/11/2026 11:31AM	\TNguyen	v-t-14-01-01.lst	03/11/2026 09:08AM	t-14-01-01-analysis.rtf	12/17/2025 11:13AM	Validation SAS is newer than validation LST.
2				v-t-14-01-01-01.sas	03/11/2026 11:55AM	\JChen				12/17/2025 11:13AM	No output. No validation LST file. No Develop Program.
3				v-t-14-01-02.sas	03/11/2026 10:57AM	\JChen	v-t-14-01-02.lst	03/11/2026 09:09AM	t-14-01-02-disp.rtf	12/17/2025 11:13AM	Validation SAS is newer than validation LST. No Develop
4											Developer and Validator are same. Need to refresh output. Validation SAS is newer than validation LST.
5	t-14-01-03.sas	01/14/2026 03:21PM	\JChen	v-t-14-01-03.sas	03/11/2026 09:10AM	\JChen	v-t-14-01-03.lst	03/11/2026 09:09AM	t-14-01-03-dv-covid.rtf	12/17/2025 11:13AM	Need to refresh output. Validation SAS is newer than validation LST.
6	t-14-01-04.sas	12/13/2023 11:13AM	\JChen								No Validation Program. No output.
7	t-14-01-05-demog.sas	03/27/2024 09:02AM	\JChen	v-t-14-01-05-demog.sas	03/11/2026 09:12AM	\JChen	v-t-14-01-05-demog.lst	03/11/2026 09:12AM	t-14-01-05-demog.rtf	12/17/2025 11:13AM	Developer and Validator are same. Developer and Validator are same.
8	t-14-01-12.sas	12/13/2023 11:13AM	\JChen	v-t-14-01-12.sas	03/11/2026 09:12AM	\JChen					No output. No validation LST file.
9	t-14-02-03-04.sas	01/14/2026 03:21PM	\JChen	v-t-14-02-03-04.sas	12/10/2025 01:33PM	\JChen	v-t-14-02-03-04.lst	02/27/2026 11:15AM			Developer and Validator are same. Developer program newer than validator's. No output.
10	t-14-10-11-sae.sas	03/16/2023 09:28AM	\JChen	v-t-14-10-11.sas	03/11/2026 09:13AM	\JChen			t-14-10-11-sae-socpt.rtf	12/17/2025 11:13AM	Developer and Validator are same. No validation LST file.

Output 1. Example of The Master Tracker Automatically Generated by the Global MACRO

LOGIC BEHIND THE GLOBAL MACRO

After user pass the delivery folder name (e.g. "%let loc=G:\Eidostx\AG10\AG10-202\Deliverable8;") to the only macro variable needed, the Global Macro will automatically go to each subfolder of sdtm, adam, tables, figures and listings under that AG10\AG10-202\Deliverable8 folder, then search all SAS programs under \develop and \validation, search all sas7bdata data sets under \data, search all RTF files under \outputs, search all LST files under \validation.

Display 2 shows the example of all table developers SAS programs. If there are other types of files (e.g. LOG, LST etc.) under the same folder, the Global Macro will ignore those files, but only pick SAS programs.

> Eidostx > AG10 > AG10-202 > Deliverable8 > tables > develop

Name	Date modified	Type	Size
t-14-01-01.sas	12/13/2023 10:13 AM	SAS System Program	6 KB
t-14-01-03.sas	1/14/2026 2:21 PM	SAS System Program	6 KB
t-14-01-04.sas	12/13/2023 10:13 AM	SAS System Program	6 KB
t-14-01-05-demog.sas	3/27/2024 9:02 AM	SAS System Program	6 KB
t-14-01-12.sas	12/13/2023 10:13 AM	SAS System Program	6 KB
t-14-02-03-04.sas	1/14/2026 2:21 PM	SAS System Program	6 KB
t-14-10-11-sae.sas	3/16/2023 9:28 AM	SAS System Program	2 KB

Display 2. Example of Developers SAS program under G:\Eidostx\AG10\AG10-202\Deliverable8\tables\develop

Display 3 shows the example of all table validators SAS programs and associated LOG and LST files. The Global Macro pick SAS programs and LST files under this folder and ignore other types of files under this folder.

1) (G:) > Eidostx > AG10 > AG10-202 > Deliverable8 > tables > validation

Name	Date modified	Type	Size
v-t-14-01-01.log	3/11/2026 9:07 AM	Text Document	3 KB
v-t-14-01-01.lst	3/11/2026 9:08 AM	SAS Output	2 KB
v-t-14-01-01.sas	3/11/2026 11:31 AM	SAS System Program	1 KB
v-t-14-01-01-01.sas	3/11/2026 11:55 AM	SAS System Program	2 KB
v-t-14-01-02.log	3/11/2026 9:09 AM	Text Document	8 KB
v-t-14-01-02.lst	3/11/2026 9:09 AM	SAS Output	2 KB
v-t-14-01-02.sas	3/11/2026 10:57 AM	SAS System Program	2 KB
v-t-14-01-03.Log	3/11/2026 9:09 AM	Text Document	1 KB
v-t-14-01-03.lst	3/11/2026 9:09 AM	SAS Output	2 KB
v-t-14-01-03.sas	3/11/2026 9:10 AM	SAS System Program	2 KB
v-t-14-01-05-demog.log	3/11/2026 9:12 AM	Text Document	1 KB
v-t-14-01-05-demog.lst	3/11/2026 9:12 AM	SAS Output	2 KB
v-t-14-01-05-demog.sas	3/11/2026 9:12 AM	SAS System Program	1 KB
v-t-14-01-12.sas	3/11/2026 9:12 AM	SAS System Program	4 KB
v-t-14-02-03-04.log	2/27/2026 10:15 AM	Text Document	27 KB
v-t-14-02-03-04.lst	2/27/2026 10:15 AM	SAS Output	23 KB
v-t-14-02-03-04.sas	12/10/2025 12:33 PM	SAS System Program	4 KB
v-t-14-10-11.Log	3/11/2026 9:12 AM	Text Document	1 KB
v-t-14-10-11.sas	3/11/2026 9:13 AM	SAS System Program	2 KB

Display 3. Example of Validators SAS program under G:\Eidostx\AG10\AG10-202\Deliverable8\tables\validation

Display 4 shows the example of all table outputs. If there are other types of files (e.g. PDF etc.) under the same folder, the Global Macro will ignore those files, but only pick RTF files.

1) (G:) > Eidostx > AG10 > AG10-202 > Deliverable8 > tables > output

Name	Date modified	Type	Size
t-14-01-01-analysis.rtf	12/17/2025 10:13 AM	Rich Text Format	5 KB
t-14-01-02-disp.rtf	12/17/2025 10:13 AM	Rich Text Format	21 KB
t-14-01-03-dv-covid.rtf	12/17/2025 10:13 AM	Rich Text Format	6 KB
t-14-01-05-demog.rtf	12/17/2025 10:13 AM	Rich Text Format	19 KB
t-14-10-11-sae-socpt.rtf	12/17/2025 10:13 AM	Rich Text Format	19 KB

Display 4. Example of Table Outputs under G:\Eidostx\AG10\AG10-202\Deliverable8\tables\output

For each SAS program, LST file, data and output, this global macro will automatically fetch developer and validator's SAS program name, validator's LST file name, output file name, when each file was generated and who is the owner. Sub macro %fetch in Program 1 realize this step.

```
%macro fetch(location=,ds=,type='.SAS ');
options symbolgen mprint mlogic;
* find all files with &type under folder &location, and associated owner and
last modified datetime, save these information to data set $ds;
filename nams pipe "dir /Q &location";

data &ds (drop=date time ap line newdt);
infile nams missover pad length=len;
input @01 line $varying200. len ;
```

```

if index(upcase(line), &type);
date=tranwrd(scan(line, 1, " "), "/", "-");
time=scan(line, 2, " ");
ap=scan(line, 3, " ");
f=scan(trim(left(line)), 6, " ");
%if &out=output %then %do; link=compress(f, "", "kd"); %end;
%else %if &out=data %then %do;
  l=index(f, ".");
  link=substr(f, 1, l-1); %end;
dt=scan(trim(left(line)), 1, " ") || " " || scan(trim(left(line)), 2, "
") || scan(trim(left(line)), 3, " ");
u=scan(trim(left(line)), 5, " ");
newdt=catx(" ", date, time, ap);
d=input(newdt, mdyampm25.);
format d dateampm25.;
run;
%mend fetch;

```

Program 1. Sub macro %fetch

Macro variable &location is the location of the deliverable folder, in this example, it is the global macro variable **loc=G:\Eidostx\AG10\AG10-202\Deliverable8**. Macro variable &ds is the output data set name. Macro variable &type is the file type we want to fetch in each subfolder.

Program 2 shows the main macro %tracker. First, it calls %fetch four times to fetch files under \develop, \validation and \&out subfolders. \&out subfolder is \data for SDTM and ADaM, it is \output for TLF.

If your company standard folder structure is different, then this is the place to slightly modify four %fetch() call to fit for your company standard. We will show how to implement it in another example later.

```

%global loc out;
%let loc=G:\Eidostx\AG10\AG10-202\Deliverable8;

%macro tracker(loc=, sub=, out=, outtype=);
* get developer SAS program name, owner and datetime of last modified;
%fetch(location=&loc&sub\develop, ds=d);
* get validator SAS program name, owner and datetime of last modified;
%fetch(location=&loc&sub\validation, ds=v);
* get validator LST program name, owner and datetime of last modified;
%fetch(location=&loc&sub\validation, ds=vl, type='.LST');
* get output file name/data set name, owner and datetime of last modified;
%fetch(location=&loc&sub&out, ds=t, type=&outtype);

data v;
  set v (rename=(f=v_f dt=v_dt u=v_u d=v_d));
  %if &out=data %then %do; link=substr(link, 3); %end;
run;
data vl;
  set vl (rename=(f=vl_f dt=vl_dt u=vl_u d=vl_d));
  %if &out=data %then %do; link=substr(link, 3); %end;
run;
data t;
  set t (rename=(f=t_f dt=t_dt u=t_u d=t_d));
run;
proc sort data=d; by link; run;
proc sort data=v; by link; run;
proc sort data=vl; by link; run;
proc sort data=t; by link; run;
data final (keep=f dt u v_f v_dt v_u vl_f vl_dt t_f t_dt issue);
  merge d (in=x) v (in=y) vl (in=a) t (in=z) ;
  by link;
  length f1-f8 $50. issue $400.;
  if ^y then f1="No Validation Program. ";
  if ^x then f8="No Develop Program. ";
  if x*y then do;
    if u=v_u then f2="Developer and Validator are same. ";
    if d>v_d then f3="Developer program newer than validator's. ";

```

```

end;
if ^z then f4="No output. ";
if x*z and d>t_d then f5="Need to refresh output. ";
if y*a and v_d>v1_d then f6="Validation SAS is newer than validation LST. ";
if y*^a then f7="No validation LST file. ";
issue=catx(" ",f1,f2,f3,f4,f5,f6,f7,f8);
label f="Develop Program Name"
      dt="Develop Program Date-time"
      u="Developer"
      v_f="Validation Program Name"
      v_dt="Validation Program Date-time"
      v_u="Validator"
      v1_f="Validation LST File Name"
      v1_dt="Validation LST Date-time"
      t_f="Output Name"
      t_dt="Output Date-time"
      t_u="Output Owner"
      issue="Issue(s) ";

run;
data final;
  set final;
run;
proc export data=final
  outfile="&loc\MasterTracker.xlsx"
  dbms=xlsx label
  replace;
  sheet="&sub";
run;
%mend tracker;

%tracker(loc=&loc, sub=sdtm, out=data, outtype='.SAS7BDAT');
%tracker(loc=&loc, sub=adam, out=data, outtype='.SAS7BDAT');
%tracker(loc=&loc, sub=tables, out=output, outtype='.RTF');
%tracker(loc=&loc, sub=figures, out=output, outtype='.RTF');
%tracker(loc=&loc, sub=listings, out=output, outtype='.RTF');

```

Program 2. Global Macro %tracker

The “link” variable is the key. For tables, figures and listings, SAS program name and output name don’t need to be exactly the same, the key is they all need to contain the same TLF number. E.g

Developer program: t-14-10-11-sae.sas

Validator program: v-t-14-10-11.sas

Output: t-14-10-11-sae-socpt.rtf

For SDTM and ADaM, almost all the company follow the standard method, e.g.:

Developer program: adsl.sas

Validator program: v-adsl.sas (or v_adsl.sas)

Output: adsl.sas7bdat

If your company use a different naming convention, then you can slightly modify the “link” variable in %tracker.

Another nice part of this global macro is, it can automatically find “issues” in the development and validation of SDTM, ADaM and TLFs. Current macro can identify 8 types of issues, please see table below. If your company is more strict or more open to these issues, you can modify derivation of F1 to F8 in the %tracker to adjust.


Variable Name in %tracker()	Issues
F1	“No Validation Program.”

F2	"Developer and Validator are same."
F3	"Developer program newer than validator's."
F4	"No output."
F5	"Need to refresh output."
F6	"Validation SAS is newer than validation LST."
F7	"No validation LST file."
F8	"No Develop Program."

If one program and/or output has multiple issues, then all triggered issues will be listed on the last column "Issue(s)" in the tracker. After those "issues" being fixed, then it will disappear in the "Issue(s)" column in next refresh. This nice function provides an automatically way to identify and resolve different issues during development and validation stage. After issues are resolved, it becomes a nice complete tracker as proof to auditors.

USE ANOTHER EXAMPLE TO DEMENSTRATE THE FLEXIBILITES OF THE GLOBAL MACRO

This example shows how easy and flexible to update this global macro based on your company's standard folder structure. Suppose your company's standard folder structure is similar as shown in Display 5. \production subfolder contains developer's SAS programs, LOG/LST and output or data. \QC subfolder contains validator's SAS programs, LOG/LST.



Display 5. Another Example of Study Folder Structure

The only change is the four call of %fetch at the beginning of of %tracker() and pass slightly different macro variables when calling %tracker() as shown in BLUE text in Program 3 below.

```
%macro tracker(loc=,sub=,out=,outtype=);
* get developer SAS program name, owner and datetime of last modified;
%fetch(location=&loc&sub\production,ds=d);
```

```

* get validator SAS program name, owner and datetime of last modified;
%fetch(location=&loc\&sub\QC,ds=v);
* get validator LST program name, owner and datetime of last modified;
%fetch(location=&loc\&sub\QC,ds=vl,type='.LST');
* get output file name/data set name, owner and datetime of last modified;
%fetch(location=&loc\&sub\outproduction,ds=t,type=&outtype);
...
...

%mend tracker;

%tracker(loc=&loc,sub=sdtm,out=data,outtype='.SAS7BDAT');
%tracker(loc=&loc,sub=adam,out=data,outtype='.SAS7BDAT');
%tracker(loc=&loc,sub=tables,out=output,outtype='.RTF');
%tracker(loc=&loc,sub=figures,out=output,outtype='.RTF');
%tracker(loc=&loc,sub=listings,out=output,outtype='.RTF');

```

Program 3. An example to demonstrate the flexibility of the Global Macro %tracker

Output 2 shows an example of ADaM tab in the master tracker automatically generated by the global macro based on the different folder structure as shown in Display 5.

	A	B	C	D	E	F	G	H	I	J	K
	Develop Program Name	Develop Program Date-time	Developer	Validation Program Name	Validation Program Date-time	Validator	Validation LST File Name	Validation LST Date-time	Output Name	Output Date-time	Issue(s)
1	adae.sas	09/13/2023 08:19PM	BRIDGEBIOSAS \JChen	v-adae.sas	03/11/2026 09:46AM	BRIDGEBIOSAS \JChen	v-adae.lst	03/11/2026 09:46AM	adae.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same.
3	adcm.sas	08/30/2023 09:31AM	BRIDGEBIOSAS \JChen						adcm.sas7bdat	11/20/2023 12:38PM	No Validation Program.
4	adeg.sas	08/09/2023 07:02AM	BRIDGEBIOSAS \JChen	v-adeq.sas	03/11/2026 09:46AM	BRIDGEBIOSAS \JChen	v-adeq.lst	03/11/2026 09:46AM	adeg.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same.
5	adex.sas	08/18/2023 10:07AM	BRIDGEBIOSAS \JChen	v-adex.sas	03/11/2026 09:44AM	BRIDGEBIOSAS \JChen	v-adex.lst	03/11/2026 09:44AM	adex.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same.
6	adlb.sas	09/14/2023 04:23PM	BRIDGEBIOSAS \JChen	v-adlb.sas	03/11/2026 09:47AM	BRIDGEBIOSAS \JChen	v-adlb.lst	03/11/2026 09:46AM	adlb.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same. Validation SAS is newer than validation LST.
7				v-adlbhys.sas	03/11/2026 09:47AM	BRIDGEBIOSAS \JChen			adlbhys.sas7bdat	11/20/2023 12:38PM	No validation LST file. No Develop Program.
8	admh.sas	08/30/2023 09:35AM	BRIDGEBIOSAS \JChen	v-admh.sas	03/11/2026 09:47AM	BRIDGEBIOSAS \JChen	v-admh.lst	03/11/2026 09:47AM	admh.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same.
9	adpc.sas	03/11/2026 09:51AM	BRIDGEBIOSAS \JChen	v-adpc.sas	12/31/2025 10:18AM	BRIDGEBIOSAS \JChen					Developer and Validator are same. Developer program newer than validator's. No output. No validation LST file.
10	adpe.sas	08/30/2023 09:26AM	BRIDGEBIOSAS \JChen	v-adpe.sas	03/11/2026 09:45AM	BRIDGEBIOSAS \JChen			adpe.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same. No validation LST file.
11	adsl.sas	09/14/2023 12:21PM	BRIDGEBIOSAS \JChen	v-adsl.sas	03/11/2026 09:47AM	BRIDGEBIOSAS \JChen	v-adsl.lst	03/11/2026 09:47AM	adsl.sas7bdat	11/20/2023 12:38PM	Developer and Validator are same.
12	adv.sas	08/16/2023 02:56PM	BRIDGEBIOSAS \JChen	v-adv.sas	03/11/2026 09:45AM	BRIDGEBIOSAS \JChen	v-adv.lst	03/11/2026 09:45AM			Developer and Validator are same. No output.

Output 1 . Another example of Master Tracker Shows ADaM Tab Based on a Different Folder Structure

CONCLUSION

The global macro described in this paper provides a simple, quick and handy solution to manage SDTM, ADaM and TLF's development and validation, automatically identify potential issues and generate a master tracker which can also be used for audit purposes.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Zhuo Chen (Joyce)
 BridgeBio Pharma, LLC
 Joyce.Chen@bridgebio.com

Ted Lystig
 BridgeBio Pharma, LLC
 Ted.Lystig@bridgebio.com

Martha Cao
BridgeBio Pharma, LLC
Martha.Cao@bridgebio.com

Tuan Nguyen
BridgeBio Pharma, LLC
Tuan.Nguyen@bridgebio.com